***                                                              ***;
***                      Marco GILARDETTI                        ***;
***                                                              ***;
***                 Hickok 6000A - Hickok I177                   ***;
***  Integrity check, clean up and tube chart conversion program ***;
***                 written for SAS engine V8                    ***;
***                                                              ***;
***           A project by M. Gilardetti and D. Spear            ***;
***                                                              ***;
libname hickok 'C:\Hickok 6000-I177';

***                                                              ***;
***                         Data import                          ***;
***                                                              ***;
/* 

NOTE: do not use SAS import procedure (will mismatch the data).

NOTE: external procedure; run this batch on DBMS/COPY:

--------------------------------------------------------------------

compute;

in=C:\Hickok 6000-I177\[a,2,2043,a,h,1,1]cumulativo.xls7

out=C:\Hickok 6000-I177\cumulativo.sas7sun;

run;

------------------------------------------------------------------*/
* Clear up data attributes;
DATA hickok;

set hickok.cumulativo;

retain TUBE_ID 0;

format _all_;

informat _all_;

attrib _all_ label='';

TUBE_ID + 1;

run;

***                                                              ***;
***                           Checkup                            ***;
***                                                              ***;
PROC contents data = hickok; RUN;

* Check filament voltage values;
DATA hickok;

set hickok;

row2 = tranwrd(row2,'.',',');

row2 = tranwrd(row2,',0','');

PROC freq data = hickok; 

title 'VOLTAGE values found in data unavailable on instrument';

table ROW2 /list missing;

where row2 not in('1,1','1,5','2','2,5','3','4,3','5','6,3','7,5','10','12,6','25',

                  '35','50','75','117','OFF');

PROC print data = hickok;

title 'VOLTAGE settings to be verified';

var row1 row2;

where row2 not in('0,6','1,1','1,4','2','2,5','3','4,3','5','6,3','7,5','10',

                  '12,6','17','20','25','35','50','75','117','OFF');

RUN;

* Check knobs position;
DATA hickok checkrow3; 

set hickok;

length check $9;

array knobs $ 1 knob1-knob9;

row3 = tranwrd(row3,'O','0');

do i = 1 to 9;


knobs(i) = substr(row3,i,1);

end;

if knob1 not in ('A','B','C','D','E','F','G','H','J','K','L','M') then do;


check = 'knob1 = ' || knob1; output checkrow3; end;

if knob2 not in ('1','2','3','R','S','T','U','V','W','X','Y','Z') then do;


check = 'knob2 = ' || knob2; output checkrow3; end;

if knob3 ne '-' then do;


check = 'knob3 = ' || knob3; output checkrow3; end;

if knob4 not in ('0','1','2','3','4','5','6','7','8','9','X','A') then do;


check = 'knob4 = ' || knob4; output checkrow3; end;

if knob5 not in ('0','1','2','3','4','5','6','7','8','9','X','A') then do;


check = 'knob5 = ' || knob5; output checkrow3; end;

if knob6 not in ('0','1','2','3','4','5','6','7','8','9','X','A') then do;


check = 'knob6 = ' || knob6; output checkrow3; end;

if knob7 not in ('0','1','2','3','4','5','6','7','8','9','X','A') then do;


check = 'knob7 = ' || knob7; output checkrow3; end;

if knob8 ne '-' then do;


check = 'knob8 = ' || knob8; output checkrow3; end;

if knob9 not in ('0','1','2','3','4','5','6','7','8','9','X','A') then do;


check = 'knob9 = ' || knob9; output checkrow3; end;

output hickok;

PROC print data = checkrow3;

title 'Knobs with wrong label, CHECK = knob to be verified';

format row1 $10. row2 $4. row3 $9. row4-row5 $3. row6 $1. row7 $5. row8 $50.;

var row1 row2 row3 check row4-row7 row8;

RUN;

* Check first potentiometer setting;
DATA hickok checkrow4; 

set hickok;

row4 = compress(row4);

if row4 ne 'Vary' then do;


if not (('0' le substr(row4,1,1) le '9') and (('0' le substr(row4,2,1) le '9') 

    or substr(row4,2,1) eq '')) and (row4 ne '100') then output checkrow4;

end;

output hickok;

PROC print data = checkrow4;

title 'BIAS settings to be verified';

format row1 $10. row4-row5 $3. row8 $50.;

var tube_id row1 row4-row5 row8;

RUN;

* Check second potentiometer setting;
DATA hickok checkrow5; 

set hickok;

row5 = compress(row5);

if not (('0' le substr(row5,1,1) le '9') and (('0' le substr(row5,2,1) le '9') 

or substr(row5,2,1) eq '')) and (row5 ne '100') then output checkrow5;

output hickok;

PROC print data = checkrow5;

title 'SHUNT settings to be verified';

format row1 $10. row4-row5 $3. row8 $50.;

var tube_id row1 row4-row5 row8;

RUN;

* Check button to be pressed;
DATA hickok checkrow6; 

set hickok;

if row6 not in ('A','B','C','D','F') then output checkrow6;

output hickok;

PROC print data = checkrow6;

title 'FUNCTION settings to be verified';

format row1 $10. row6 $1. row8 $50.;

var tube_id row1 row6 row7 row8;

RUN;

* Check micromhos meter reading;
DATA hickok checkrow7; 

set hickok;

if verify(compress(row7),'0123456789') ne 0 or row7 eq '' or length(row7) gt 5 then do;


output checkrow7;


row7 = '';

end;

row7 = compress(row7);

output hickok;

PROC freq data=checkrow7; 

title 'METER readings to be verified';

tables row7 /missing;

PROC print data = checkrow7;

format row1 $10. row7 $10. row8 $50.;

var row1 row7 row8;

where verify(compress(row7),'01232456789-_') ne 0;

RUN;

***                                                              ***;
***                        Data conversion                       ***;
***                                                              ***;
* Outputs data for models 6000/6000A/6005;
DATA hickok_i177;

set hickok;

row1 = tranwrd(row1,'!','');

DATA hickok.hickok_6000_6005;

set hickok;

row1 = tranwrd(row1,'!','§');

keep row1-row8;

rename row1=TUBE_TYPE row2=FIL row3=SELECTORS row4=BIAS row5=SHUNT


   row6=FUNCTION row7=MUT_COND row8=NOTATIONS;

drop tube_id;

PROC sort data=hickok.hickok_6000_6005;

by tube_type;

RUN;

* Clean up tube names;
DATA hickok_i177;

set hickok;

row1 = tranwrd(row1,'!','');

RUN;

* Translate voltage values;
DATA hickok_i177;

set hickok_i177;

length VOLT $4 no177a 3;

volt = compress(row2);

if volt in ('0,6','17','20') then no177a = 1;

else if volt eq '1,4' then volt = '1,5';

RUN;

* Translate knobs positions;
DATA hickok_i177;

set hickok_i177;

length newknob1-newknob7 $2 JACK1-JACK7 $8 no177b no177d 3;

if row1 eq '8552' then no177d = 1;

if knob6 eq 'W' then do; knob6 = ''; no177d = 1; end; * Typo?;
newknob1 = translate(knob1,'0123456789','ABCDEFGHJK'); * L and M are never used;
if knob1 eq '1' then do; newknob1 = '10'; no177c = 1; end; * Nuvistors;
newknob2 = translate(knob2,'123456789','RSTUVWXYZ'); * 1 and 2 are never used;
if knob2 = '3' then newknob2 = '12'; * Compactrons only;
newknob3 = tranwrd(knob4,'X','10'); newknob3 = tranwrd(newknob3,'A','11');

newknob4 = tranwrd(knob5,'X','10'); newknob4 = tranwrd(newknob4,'A','11');

newknob5 = tranwrd(knob6,'X','10'); newknob5 = tranwrd(newknob5,'A','11');

newknob6 = tranwrd(knob7,'X','10'); newknob6 = tranwrd(newknob6,'A','11');

newknob7 = tranwrd(knob9,'X','10'); newknob7 = tranwrd(newknob7,'A','11');

if newknob1 ne '0' then jack1 = trim('FIL =')||' '||newknob1;

if newknob2 ne '0' then do;


if newknob2 eq 'P' and row6 ne 'G' then do; no177b = 1; jack2 = 'FIL ='; end;


else if newknob2 eq 'P' and row6 eq 'G' then do; jack2 = ''; end;


else jack2 = trim('FIL =')||' '||newknob2;

end;

if newknob3 ne '0' then jack3 = trim('G =')||' '||newknob3;

if newknob4 ne '0' then jack4 = trim('P =')||' '||newknob4;

if newknob5 ne '0' then jack5 = trim('SC =')||' '||newknob5;

if newknob6 ne '0' then jack6 = trim('K =')||' '||newknob6;

if newknob7 ne '0' then jack7 = trim('K =')||' '||newknob7;

if newknob3 eq 'G = 10' then newknob3 = 'G=10(CAP)'; * as a reminder;
if newknob4 eq 'P = 10' then newknob3 = 'P=10(CAP)'; * as a reminder;
RUN;

* Translate 'press' button;
DATA hickok_i177;

set hickok_i177;

length PRESS $14;

if row6 not in ('A','C','D','E','G',' ') then do;


no177b = 1;


press = trim(row6)||' (SEE NOTES)';

end;

if row6 eq 'A' then press = 'AMPL. TEST';

else if row6 eq 'C' then press = 'DIODE TEST';

else if row6 eq 'D' then press = 'RECT. ST''D';

else if row6 eq 'E' then press = 'RECTIF. 117N7';

else if row6 eq 'G' then press = 'OZ4 BUTTON';

else if row6 eq ' ' then press = '(CHECK SHORTS)';

RUN;

* Translate BIAS and SHUNT into L and R;
DATA hickok_i177;

set hickok_i177;

length temp_L temp_R 8 L R $4;

temp_L = input(row4,?? best5.);

temp_R = input(row5,?? best5.);

temp_L = round((temp_L/100)*80);

temp_R = round((temp_R/100)*80);

R = compress(put(temp_L,4.0)); * Values are swapped;
L = compress(put(temp_R,4.0)); * Values are swapped;
if L eq '.' then L = row5; * Values are swapped;
if R eq '.' then R = row4; * Values are swapped;
RUN;

* Add a full stop at the very end of NOTATIONS field;
DATA hickok_i177;

set hickok_i177;

if row8 ne '' then row8 = trim(row8)||'.';

row8 = tranwrd(row8,'..','.');

RUN;

* Add supplementary notes when necessary;
DATA hickok_i177;

set hickok_i177;

length NOTATIONS $330;

notations = trim(left(row8));

if no177a eq 1 then notations = trim(notations)||

        ' SPECIFIED FILAMENT VOLTAGE NOT AVAILABLE ON MODEL I-177; USE EXTERNAL SUPPLY.';

if no177b eq 1 then notations = trim(notations)||



' TEST PROCEDURE UNKNOWN WITH MODEL I-177. DO NOT TEST.';

if no177c eq 1 then notations = trim(notations)||



' MODEL I-177: USE THIS SETTING (CHECK PINOUT ON DATA CHART).';

if no177d eq 1 then notations = trim(notations)||

        ' MODEL I-177: PIN WIRING INCOMPLETE, MUST BE DESUMED ELSEWHERE.';

notations = upcase(left(notations));

notes_length = length(notations);

RUN;

* Outputs data for model I-177;
DATA hickok.hickok_i177;

set hickok_i177;

format row1 $8. row2 $4. row7 $5.;

keep row1 volt row7 jack1-jack7 press L R notations;

rename row1=TUBE_TYPE row7=AVG_MC

       jack1=FIL1 jack2=FIL2 jack3=GRID jack4=PLATE jack5=SCREEN jack6=CATH jack7=SUPPR;

RUN;

PROC sort data=hickok.hickok_i177;

by tube_type volt;

RUN;
